
1) Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 6.4.7, which is part of the LTE test suite. 

The test case can be demonstrated to run with one LTE UE. Execution log files are provided as evidence. 

3. Verification Test Summary

Test Case:
TC_6_4_7
Test Group:
Hybrid Cells
ATS Version:
D16wk10
System Simulator used:
Anritsu Protocol Conformance Test System ME7834L
UE used:
INTEL 7360
Verification Status:
PASS


4. Corrections required for TC 6.4.7
Change 1 – Correction to function ‘testcase TC_6_4_7()
	Function name
	testcase TC_6_4_7()

	Reason for change
	Test case fails intermittently as a result of the 6 minute Guard timer time out at step 31. 
Note:- See table 6.4.7.3.2-3 per prose completion of step 31 may up to 6 min.

	Summary of change
	Extend Guard timer from 6 min to 9 min.

	TTCN module
	LTE_IRAT\source\LTE_IRAT_Testsuite.ttcn


Before change

  testcase TC_6_4_7() runs on MTC_LTE_IRAT system SYSTEM_LTE_IRAT {

    // @purpose

    //   Inter-RAT cell reselection / From GERAN to E-UTRA RRC_IDLE member hybrid cell

    var EUTRA_PTC        v_EUTRA      := null;

    var UTRAN_IRAT_PTC   v_UTRAN      := null;

    var GERAN_PTC        v_GERAN      := null;

    var CDMA2000_PTC     v_CDMA2000   := null;

    var IMS_PTC          v_IMS1       := null;

    var IMS_PTC          v_IMS2       := null;

    // Anritsu_CSG_647 - 6min is not long enough. Cos step 31 take 6 min already

    timer t_GuardTimer := int2float(360);
    v_EUTRA := EUTRA_PTC.create alive;

    v_GERAN := GERAN_PTC.create alive;

    f_MTC_ConnectPTCs_LTE_IRAT(system, v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000, v_IMS1, v_IMS2);

    v_EUTRA.start(f_TC_6_4_7_EUTRA());

    v_GERAN.start(f_TC_6_4_7_GERAN());

    t_GuardTimer.start;

    f_MTC_MainLoop_LTE_IRAT(t_GuardTimer);

  }
After change

  testcase TC_6_4_7() runs on MTC_LTE_IRAT system SYSTEM_LTE_IRAT {

    // @purpose

    //   Inter-RAT cell reselection / From GERAN to E-UTRA RRC_IDLE member hybrid cell

    var EUTRA_PTC        v_EUTRA      := null;

    var UTRAN_IRAT_PTC   v_UTRAN      := null;

    var GERAN_PTC        v_GERAN      := null;

    var CDMA2000_PTC     v_CDMA2000   := null;

    var IMS_PTC          v_IMS1       := null;

    var IMS_PTC          v_IMS2       := null;

    // Anritsu_CSG_647 – Change 1.1 - 6min is not long enough. Cos step 31 take 6 min already
    timer t_GuardTimer := int2float(540);  

    v_EUTRA := EUTRA_PTC.create alive;

    v_GERAN := GERAN_PTC.create alive;

    f_MTC_ConnectPTCs_LTE_IRAT(system, v_EUTRA, v_UTRAN, v_GERAN, v_CDMA2000, v_IMS1, v_IMS2);

    v_EUTRA.start(f_TC_6_4_7_EUTRA());

    v_GERAN.start(f_TC_6_4_7_GERAN());

    t_GuardTimer.start;

    f_MTC_MainLoop_LTE_IRAT(t_GuardTimer);

  }  
Change 2 – Correction to function ‘f_EUTRA_CellInfo_SetSIB4’ 
	Function name
	f_EUTRA_CellInfo_SetSIB4

	Reason for change
	Missing csg_PhysCellIdRange info IEs in SIB4 per prose Table 6.4.7.3.3-2  

	Summary of change
	Add combination c14 handling to function call f_EUTRA_CellInfo_SetSIB4

	TTCN module
	Common\EUTRA\EUTRA_CellInfo.ttcn


Before change

  function f_EUTRA_CellInfo_SetSIB4(EUTRA_CellId_Type p_CellId,

                                    template (value) SystemInformationBlockType4 p_Sib4) runs on EUTRA_PTC

  {

    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get(p_CellId);

    // SIB4 is only present in combination c2 and c7 (see 36.508 cl. 4.4.3 Table 4.4.3.1.1-1 for more information)

    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c2, c7, c11, c13) {   // @sic R5s110166 sic@ 

       //The combination initiation can be found in f_EUTRA_InitSystemInformation

        v_CellInfo.Sysinfo.BCCH_Info.SIs[2].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib4 := p_Sib4;

        f_EUTRA_CellInfo_Set(p_CellId, v_CellInfo);

      }

    }

  }
After change

  function f_EUTRA_CellInfo_SetSIB4(EUTRA_CellId_Type p_CellId,

                                    template (value) SystemInformationBlockType4 p_Sib4) runs on EUTRA_PTC

  {

    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get(p_CellId);

    // SIB4 is only present in combination c2 and c7 (see 36.508 cl. 4.4.3 Table 4.4.3.1.1-1 for more information)

    select (valueof(v_CellInfo.Sysinfo.Combination)) {

      // Anritsu_CSG_632_647

     case (c2, c7, c11, c13, c14) {   // @sic R5s110166 sic@ 

        //The combination initiation can be found in f_EUTRA_InitSystemInformation

        v_CellInfo.Sysinfo.BCCH_Info.SIs[2].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib4 := p_Sib4;

        f_EUTRA_CellInfo_Set(p_CellId, v_CellInfo);

      }

    }

  }
Change 3 – Correction to function ‘f_EUTRA_CellInfo_SetSIB7’ 
	Function name
	function f_EUTRA_CellInfo_SetSIB7

	Reason for change
	Missing csg_PhysCellIdRange info IEs in SIB4 per prose Table 6.4.6.3.3-4  

	Summary of change
	Add combination c14 handling to function call f_EUTRA_CellInfo_SetSIB7

	TTCN module
	Common\EUTRA\EUTRA_CellInfo.ttcn


Before change

  function f_EUTRA_CellInfo_SetSIB7(EUTRA_CellId_Type p_CellId,

                                    template (value) SystemInformationBlockType7 p_SIB7) runs on EUTRA_PTC

  {

    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get(p_CellId);

    select (valueof(v_CellInfo.Sysinfo.Combination)) {

      case (c5) {

        v_CellInfo.Sysinfo.BCCH_Info.SIs[2].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib7 := p_SIB7;

        f_EUTRA_CellInfo_Set (p_CellId, v_CellInfo);

      }
      case (c10, c10a, c12) {  // @sic R5-106551 sic@

        v_CellInfo.Sysinfo.BCCH_Info.SIs[3].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib7 := p_SIB7;

        f_EUTRA_CellInfo_Set (p_CellId, v_CellInfo);

      }

      case else {}  // Other combinations: no change needed

    }

  }
After change

  function f_EUTRA_CellInfo_SetSIB7(EUTRA_CellId_Type p_CellId,

                                    template (value) SystemInformationBlockType7 p_SIB7) runs on EUTRA_PTC

  {

    var template (value) EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get(p_CellId);

    select (valueof(v_CellInfo.Sysinfo.Combination)) {

      case (c5) {

        v_CellInfo.Sysinfo.BCCH_Info.SIs[2].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib7 := p_SIB7;

        f_EUTRA_CellInfo_Set (p_CellId, v_CellInfo);

      }

      // Anritsu_CSG_632_647 - add c14

      case (c10, c10a, c12, c14) {  // @sic R5-106551 sic@

        v_CellInfo.Sysinfo.BCCH_Info.SIs[3].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib7 := p_SIB7;

        f_EUTRA_CellInfo_Set (p_CellId, v_CellInfo);

      }

      case else {}  // Other combinations: no change needed

    }

  }
Change 4 – Correction to function
	Function name
	function f_TC_6_4_7_GERAN

	Reason for change
	1) Missing  GERAN priority Info. IEs in 2quater per per prose Table 6.4.7.3.3-3 
2) Missing co-ordination message from EUTRA_PTC in the preamble.

	Summary of change
	1)  Add  GERAN priority Info. IEs accordingly
2) Add function call f_IRAT_WaitForCoOrd_Trigger.

	TTCN module
	LTE_IRAT\6_4\Hybrid_CellsUG_GERAN.ttcn


Before change

function f_TC_6_4_7_GERAN ( ) runs on GERAN_PTC

  {//Inter-RAT cell reselection / From GSM_Idle/GPRS Packet_Idle to E-UTRA RRC_IDLE member hybrid cell

    var integer v_RF_SignalLevel_Minus80 := -80;

    var integer v_RF_SignalLevel_Minus90 := -90;

    var integer v_RF_SignalLevel_Off := tsc_GERAN_ChPwrLvl_Off;

    var IRAT_Coordination_MSG v_OctetData;

    //Initialise all cells

    f_GERAN_Init (EUTRA_GERAN);

    //Set RAC of cell 25 to different from RAC of cell 24 according to Pre-test condition: "Cell 5 and Cell 7 have different Routing Area."

    f_GERAN_RAC_Set(geran_Cell25, '02'O);

    //Create GERAN cells

    f_GERAN_CreateCell (geran_Cell24);

    f_GERAN_CreateCell (geran_Cell25);

    //Coordinate with E-UTRA component to send GERAN frequencies

    f_GERAN_Send_IR_SysInfoToEUTRA ( geran_Cell24 );

    f_GERAN_Send_IR_SysInfoToEUTRA ( geran_Cell25 );

    f_GERAN_TestBody_Set( true );    //Set power levels according to T0

    f_GERAN_SetCellPowerLevel(geran_Cell24, v_RF_SignalLevel_Minus80);

    //@siclog "Step 1" siclog@

    //The UE is switched on.

    f_UT_SwitchOnUE(UT, false);

    //@siclog "Step 2" siclog@

    //TS 34.123-3 subclause 6.10.2.7.1.1 performed on Cell 24

    //NOTE: The UE performs registration and the RR connection is released.

    f_GERAN_IdleUpdated (geran_Cell24);

    //Hand over test case to EUTRA PTC by sending auth params

    f_GERAN_SendAuthParameters (EUTRA);

    //Wait for trigger from EUTRA PTC to continue the test case

    //The trigger is updated auth params

    f_GERAN_InterRAT_InitialiseAuthParams();

    //@siclog "Step 6" siclog@

    //The SS changes Cell 1, Cell 2, Cell 24 and Cell 25 levels according to row "T2" in table 6.4.7.3.2-1 and table 6.4.7.3.2-2.

    f_GERAN_SetCellPowerLevel(geran_Cell24, v_RF_SignalLevel_Off);

    f_GERAN_SetCellPowerLevel(geran_Cell25, v_RF_SignalLevel_Minus80);

    //@siclog "Step 7-17" siclog@

    //does the test result of generic test procedure in 36.508

    //subclause 6.4.2.9 indicate that the UE is camped on GERAN Cell 25?

    EUTRA.receive (cmr_IRAT_OctetData) -> value v_OctetData; // @sic R5s130883 sic@

    f_GERAN_IdleMode_RAU( geran_Cell25, true, oct2int(v_OctetData.OctetData) ); // @sic R5s130050, R5s130883 sic@

    //@siclog "Step 18" siclog@

    //The SS changes Cell 1, Cell 24 and Cell 25 levels according to row "T3" in table 6.4.7.3.2-1 and table 6.4.7.3.2-2.

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_SetCellPowerLevel(geran_Cell24, v_RF_SignalLevel_Minus80);

    f_GERAN_SetCellPowerLevel(geran_Cell25, v_RF_SignalLevel_Minus90);

    //@siclog "Step 19-29" siclog@

    f_GERAN_IdleMode_RAU( geran_Cell24, true );

    //@siclog "Step 30" siclog@

    //The SS changes Cell 1, Cell 2 and Cell 25 level according to row "T4" in table 6.4.7.3.2-1 and table 6.4.7.3.2-2.

    f_GERAN_SendAuthParameters (EUTRA);

    f_GERAN_SetCellPowerLevel(geran_Cell25, v_RF_SignalLevel_Off);

    //Wait for end of test case

    f_IRAT_WaitForCoOrd_Trigger (EUTRA);

    f_GERAN_TestBody_Set (false);

    //Release GERAN Cells

    f_GERAN_SS_CellRelease (geran_Cell24);

    f_GERAN_SS_CellRelease (geran_Cell25);

  }

}
After change

  function f_TC_6_4_7_GERAN ( ) runs on GERAN_PTC

  {//Inter-RAT cell reselection / From GSM_Idle/GPRS Packet_Idle to E-UTRA RRC_IDLE member hybrid cell

    var integer v_RF_SignalLevel_Minus80 := -80;

    var integer v_RF_SignalLevel_Minus90 := -90;

    var integer v_RF_SignalLevel_Off := tsc_GERAN_ChPwrLvl_Off;

    var IRAT_Coordination_MSG v_OctetData;

    // Anritsu_CSG_647 – Change 3.1
    var template(value) SYSTEMINFORMATIONTYPE2quater v_SI2quater1, v_SI2quater2;
    //Initialise all cells

    f_GERAN_Init (EUTRA_GERAN);

    //Set RAC of cell 25 to different from RAC of cell 24 according to Pre-test condition: "Cell 5 and Cell 7 have different Routing Area."

    f_GERAN_RAC_Set(geran_Cell25, '02'O);

    // Anritsu_CSG_647 – Change 3.1

    v_SI2quater1 := f_GERAN_SysInfo_GetSI2q_1 (geran_Cell24);

    v_SI2quater1.sI2quaterRO.geranPrior := '101'B;

    f_GERAN_SysInfo_SetSI2q_1 (geran_Cell24, v_SI2quater1);

    v_SI2quater2 := f_GERAN_SysInfo_GetSI2q_2 (geran_Cell24);

    if (isvalue (v_SI2quater2)) {

      v_SI2quater2.sI2quaterRO.geranPrior := '101'B;

      f_GERAN_SysInfo_SetSI2q_2 (geran_Cell24, v_SI2quater2);

    }

    v_SI2quater1 := f_GERAN_SysInfo_GetSI2q_1 (geran_Cell25);

    v_SI2quater1.sI2quaterRO.geranPrior := '101'B;

    f_GERAN_SysInfo_SetSI2q_1 (geran_Cell25, v_SI2quater1);

    if (isvalue (v_SI2quater2)) {

      v_SI2quater2.sI2quaterRO.geranPrior := '101'B;

      f_GERAN_SysInfo_SetSI2q_2 (geran_Cell25, v_SI2quater2);

    }      

    //Create GERAN cells

    f_GERAN_CreateCell (geran_Cell24);

    f_GERAN_CreateCell (geran_Cell25);

    //Coordinate with E-UTRA component to send GERAN frequencies

    f_GERAN_Send_IR_SysInfoToEUTRA ( geran_Cell24 );

    f_GERAN_Send_IR_SysInfoToEUTRA ( geran_Cell25 );

    // Anritsu_CSG_647 – Change 3.2
    f_IRAT_WaitForCoOrd_Trigger (EUTRA);
    f_GERAN_TestBody_Set( true );    //Set power levels according to T0

    f_GERAN_SetCellPowerLevel(geran_Cell24, v_RF_SignalLevel_Minus80);

    //@siclog "Step 1" siclog@

    //The UE is switched on.

    f_UT_SwitchOnUE(UT, false);

    //@siclog "Step 2" siclog@

    //TS 34.123-3 subclause 6.10.2.7.1.1 performed on Cell 24

    //NOTE: The UE performs registration and the RR connection is released.

    f_GERAN_IdleUpdated (geran_Cell24);

    //Hand over test case to EUTRA PTC by sending auth params

    f_GERAN_SendAuthParameters (EUTRA);

    //Wait for trigger from EUTRA PTC to continue the test case

    //The trigger is updated auth params

    f_GERAN_InterRAT_InitialiseAuthParams();

    //@siclog "Step 6" siclog@

    //The SS changes Cell 1, Cell 2, Cell 24 and Cell 25 levels according to row "T2" in table 6.4.7.3.2-1 and table 6.4.7.3.2-2.

    f_GERAN_SetCellPowerLevel(geran_Cell24, v_RF_SignalLevel_Off);

    f_GERAN_SetCellPowerLevel(geran_Cell25, v_RF_SignalLevel_Minus80);

    //@siclog "Step 7-17" siclog@

    //does the test result of generic test procedure in 36.508

    //subclause 6.4.2.9 indicate that the UE is camped on GERAN Cell 25?

    EUTRA.receive (cmr_IRAT_OctetData) -> value v_OctetData; // @sic R5s130883 sic@

    f_GERAN_IdleMode_RAU( geran_Cell25, true, oct2int(v_OctetData.OctetData) ); // @sic R5s130050, R5s130883 sic@

    //@siclog "Step 18" siclog@

    //The SS changes Cell 1, Cell 24 and Cell 25 levels according to row "T3" in table 6.4.7.3.2-1 and table 6.4.7.3.2-2.

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_SetCellPowerLevel(geran_Cell24, v_RF_SignalLevel_Minus80);

    f_GERAN_SetCellPowerLevel(geran_Cell25, v_RF_SignalLevel_Minus90);

    //@siclog "Step 19-29" siclog@

    f_GERAN_IdleMode_RAU( geran_Cell24, true );

    //@siclog "Step 30" siclog@

    //The SS changes Cell 1, Cell 2 and Cell 25 level according to row "T4" in table 6.4.7.3.2-1 and table 6.4.7.3.2-2.

    f_GERAN_SendAuthParameters (EUTRA);

    f_GERAN_SetCellPowerLevel(geran_Cell25, v_RF_SignalLevel_Off);

    //Wait for end of test case

    f_IRAT_WaitForCoOrd_Trigger (EUTRA);

    f_GERAN_TestBody_Set (false);

    //Release GERAN Cells

    f_GERAN_SS_CellRelease (geran_Cell24);

    f_GERAN_SS_CellRelease (geran_Cell25);

  }

}
Change 5 – Correction to function ‘function f_TC_6_4_7_EUTRA’ 
	Function name
	function f_TC_6_4_7_EUTRA

	Reason for change
	1) Inocrrect initialization SIB 7 per prose Table 6.4.7.3.3-4. 

2) Missing co-oridination message to GERAN PTC in the preamble
3) Test step 5 function call f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT contains incorrect p_PreambleOnLTE and this is needed because preamble executed in GERAN.
4) Test step 31 function call f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT contains incorrect p_PreambleOnLTE and this is needed because preamble executed in GERAN.
5) Simplify the postamble using available function call.

	Summary of change
	1) Correct the SIB7 initialization as appropriate.
2) Add f_IRAT_SendCoOrd (GERAN, cms_IRAT_Trigger).

3) Replace incorrect p_PreambleOnLTE parameter in function call f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT with FALSE.

4) Replace incorrect p_PreambleOnLTE parameter in function call f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT with FALSE.
5) Replace function calls f_EUTRA_UE_Detach_SwitchOffUe, f_EUTRA_EmptyACSGL and f_EUTRA_ReleaseAllCells with a single call f_EUTRA_Postamble_EmptyCSG_AllowedList.

	TTCN module
	LTE_IRAT\6_4\Hybrid_CellsUG.ttcn


Before change

  function f_TC_6_4_7_EUTRA ( ) runs on EUTRA_PTC {

  //Inter-RAT cell reselection / From GSM_Idle/GPRS Packet_Idle to E-UTRA RRC_IDLE member hybrid cell

    var GutiParameters_Type v_GutiParams;

    var NAS_PlmnId v_PLMN;

    var template (value) CellPowerList_Type v_CellPowerList_Preamble,

                                            v_CellPowerList_T0,

                                            v_CellPowerList_T0toT1,

                                            v_CellPowerList_T1toT2,

                                            v_CellPowerList_T2toT3,

                                            v_CellPowerList_T3toT4,

                                            v_CellPowerList_Postamble;

    var CSG_Identity v_CSG_Identity_Cell2 := '000000000000000000000000001'B;

    var Q_Rxlevmin_Type v_Q_Rxlevmin := -53;

    var IRAT_CoOrd_SysInfo_Type v_GERANSysInfo24, v_GERANSysInfo25;

    var SystemInformationBlockType7 v_SystemInformationBlockType7;

    var boolean v_CSInOtherRAT;

    v_CellPowerList_Preamble := {

      cs_CellPower (eutra_Cell1, tsc_NonSuitableCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T0 := {

      cs_CellPower (eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_T0toT1 := {

      cs_CellPower (eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T1toT2 := {

      cs_CellPower (eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_T2toT3 := {

      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T3toT4 := {

      cs_CellPower (eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_Postamble := {   //@sic R5-155978 sic@

      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    //Initialise all cells, security and mobile parameters

    f_EUTRA_Init(c14);

    //Set system information combination 5 in cell 1

    f_EUTRA_ChangeSysinfoCombination (eutra_Cell1, c5);

    //Coordinate with GERAN component to receive GERAN parameters

    v_GERANSysInfo24 := f_IRAT_WaitForCoOrd_SysInfo (GERAN);

    v_GERANSysInfo25 := f_IRAT_WaitForCoOrd_SysInfo (GERAN);

    //Set Sys. info for Cell 2 according to Table 6.4.7.3.3-1

    f_EUTRA_CellInfo_SetSysInfo_Csg_Ind(eutra_Cell2, false);

    f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell2, v_CSG_Identity_Cell2);

    f_EUTRA_CellInfo_SetSysInfo_Q_Rxlevmin (eutra_Cell1, v_Q_Rxlevmin);

    f_EUTRA_CellInfo_SetSysInfo_Q_Rxlevmin (eutra_Cell2, v_Q_Rxlevmin);

    //Set SIB4 for cell 2 according to Table 6.4.7.3.3-2 and according to default message contents defined in 36.508

    f_EUTRA_CellInfo_SetSIB4 (eutra_Cell2, cs_SIB4_AllParameters (omit, omit, cs_PhysCellIdRange(4, omit)));

    // Anritsu_CSG_647 – Change 1
    // Set SIB7 for Cells 1 & 2

    //Get SIB7... @sic R5-120604 sic@

    v_SystemInformationBlockType7 := f_EUTRA_CellInfo_GetSib7(eutra_Cell1);

    // update the T Reselection..

    v_SystemInformationBlockType7.t_ReselectionGERAN := 7;

    //...and update carrierFreqsInfoList for cell 24...

    v_SystemInformationBlockType7.carrierFreqsInfoList[0].commonInfo.cellReselectionPriority := 5;

    //...and update carrierFreqsInfoList for cell 25...

    v_SystemInformationBlockType7.carrierFreqsInfoList[1].commonInfo.cellReselectionPriority := 5;

    //...the save back the SIB7

    f_EUTRA_CellInfo_SetSIB7(eutra_Cell1,

                             v_SystemInformationBlockType7);

     f_EUTRA_CellInfo_SetSIB7(eutra_Cell2,

                             v_SystemInformationBlockType7);
    //Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    // Add the CSG ID of Cell 2 in CSG list in USIM

    // Preamble

    if (pc_Manual_CSG_Selection) { //@sic R5-134565 sic@

        // Insert a USIM

        f_UT_USIM_Insert (UT, "USIM with service n86 available");

        // Empty UE Allowed CSG list in USIM

        f_EUTRA_CellInfo_SetSysInfo_Csg_Ind(eutra_Cell2, true);

        f_EUTRA_EmptyACSGL ( eutra_Cell2, {v_CSG_Identity_Cell2}); //@sic R5-155978 sic@

        // Broadcast correct CSG id

        f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell2, v_CSG_Identity_Cell2);

        f_EUTRA_SS_ConfigureSysinfo(eutra_Cell2);

        // Add CSG2 in the Alllowed CSG list in USIM

        f_EUTRA_Manual_CSG_Selection( eutra_Cell2, v_CSG_Identity_Cell2);

    } else {

        // Insert a USIM

        f_UT_USIM_Insert (UT, "USIM with service n86 available and UE CSG whitelist contains the CSG ID of Cell 2");

    }

    // Set Cell2 to Hybrid cell

    f_EUTRA_CellInfo_SetSysInfo_Csg_Ind(eutra_Cell2, false);

    f_EUTRA_SS_ConfigureSysinfo(eutra_Cell2);

    f_EUTRA_SetCellPowerList (v_CellPowerList_T0);

    f_EUTRA_TestBody_Set (true);

    //wait for GERAN PTC to perform steps 1 and 2

    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(GERAN);

    //@siclog "Step 3" siclog@

    f_EUTRA_SetCellPowerList (v_CellPowerList_T0toT1);

    //@siclog "Step 4" siclog@

    // The UE is made to perform manual CSG ID selection and select Cell 2.

    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell2);

    v_PLMN  := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    f_UT_ManualPLMN_CSG ( UT, v_PLMN, v_CSG_Identity_Cell2);

    //@siclog "Step 5" siclog@

    //Check: Does the test result of generic test procedure in TS 36.508 Table 6.4.2.7A-2 indicate that

    //the UE is camped on E-GERAN Cell 2?

    //NOTE: The UE performs a TAU procedure and the RRC connection is released.

    f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT (eutra_Cell2, v_CSInOtherRAT, false, true, omit);
    //@siclog "Step 6" siclog@

    //The SS changes Cell 1, Cell 2, Cell 24 and Cell 25 levels according

    // to row "T2" in table 6.4.7.3.2-1 and table 6.4.7.3.2-2.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T1toT2);

    f_EUTRA_SendAuthParameters(eutra_Cell2, GERAN);

    f_IRAT_SendCoOrd (GERAN, cms_IRAT_OctetData (hex2oct(tsc_EpsDefaultBearerId))); // @sic R5s130883 sic@

    f_IRAT_WaitForCoOrd_Trigger (GERAN);

    //@siclog "Step 18" siclog@

    f_EUTRA_SetCellPowerList (v_CellPowerList_T2toT3);

    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(GERAN);

    //@siclog "Step 30" siclog@

    f_EUTRA_SetCellPowerList (v_CellPowerList_T3toT4);

    //@siclog "Step 31" siclog@

    //Check that UE camps on Cell 2 with the generic test procedure TS 36.508 table 6.4.2.7A-2

    f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT (eutra_Cell2, v_CSInOtherRAT, false, true, omit);
    f_EUTRA_TestBody_Set (false);

    // Tell GERAN to cells down as test now finished

    f_IRAT_SendCoOrd (GERAN, cms_IRAT_Trigger);

    //@sic R5-155978 sic@

    f_EUTRA_UE_Detach_SwitchOffUe(eutra_Cell2, RRC_IDLE, NORMAL, omit);

    f_EUTRA_SetCellPowerList (v_CellPowerList_Postamble);

    f_EUTRA_EmptyACSGL ( eutra_Cell2, {v_CSG_Identity_Cell2});

    f_EUTRA_ReleaseAllCells();
  } // end of 6.4.7
After change

  function f_TC_6_4_7_EUTRA ( ) runs on EUTRA_PTC {

  //Inter-RAT cell reselection / From GSM_Idle/GPRS Packet_Idle to E-UTRA RRC_IDLE member hybrid cell

    var GutiParameters_Type v_GutiParams;

    var NAS_PlmnId v_PLMN;

    var template (value) CellPowerList_Type v_CellPowerList_Preamble,

                                            v_CellPowerList_T0,

                                            v_CellPowerList_T0toT1,

                                            v_CellPowerList_T1toT2,

                                            v_CellPowerList_T2toT3,

                                            v_CellPowerList_T3toT4,

                                            v_CellPowerList_Postamble;

    var CSG_Identity v_CSG_Identity_Cell2 := '000000000000000000000000001'B;

    var Q_Rxlevmin_Type v_Q_Rxlevmin := -53;

    var IRAT_CoOrd_SysInfo_Type v_GERANSysInfo24, v_GERANSysInfo25;

    var SystemInformationBlockType7 v_SystemInformationBlockType7;

    var boolean v_CSInOtherRAT;

    // Anrtitsu_CSG_647 – Change 4.1
    var template (value) CarrierFreqsInfoListGERAN v_CarrierFreqsInfoListGERAN;
    v_CellPowerList_Preamble := {

      cs_CellPower (eutra_Cell1, tsc_NonSuitableCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T0 := {

      cs_CellPower (eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_T0toT1 := {

      cs_CellPower (eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T1toT2 := {

      cs_CellPower (eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_T2toT3 := {

      cs_CellPower (eutra_Cell1, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T3toT4 := {

      cs_CellPower (eutra_Cell1, tsc_Suitable_NeighbourIntraFreq_CellRS_EPRE),

      cs_CellPower (eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_Postamble := {   //@sic R5-155978 sic@

      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    //Initialise all cells, security and mobile parameters

    f_EUTRA_Init(c14);

    //Set system information combination 5 in cell 1

    f_EUTRA_ChangeSysinfoCombination (eutra_Cell1, c5);

    //Coordinate with GERAN component to receive GERAN parameters

    v_GERANSysInfo24 := f_IRAT_WaitForCoOrd_SysInfo (GERAN);

    v_GERANSysInfo25 := f_IRAT_WaitForCoOrd_SysInfo (GERAN);

    //Set Sys. info for Cell 2 according to Table 6.4.7.3.3-1

    f_EUTRA_CellInfo_SetSysInfo_Csg_Ind(eutra_Cell2, false);

    f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell2, v_CSG_Identity_Cell2);

    f_EUTRA_CellInfo_SetSysInfo_Q_Rxlevmin (eutra_Cell1, v_Q_Rxlevmin);

    f_EUTRA_CellInfo_SetSysInfo_Q_Rxlevmin (eutra_Cell2, v_Q_Rxlevmin);

    //Set SIB4 for cell 2 according to Table 6.4.7.3.3-2 and according to default message contents defined in 36.508

    f_EUTRA_CellInfo_SetSIB4 (eutra_Cell2, cs_SIB4_AllParameters (omit, omit, cs_PhysCellIdRange(4, omit)));

    // Anritsu_CSG_647 – Change 4.1
    v_CarrierFreqsInfoListGERAN := {

      cs_CarrierFreqsInfoGERAN(cs_CarrierFreqsGERAN(v_GERANSysInfo24.Geran[0].Arfcn, f_ConvertGERAN_BandToBandIndicatorGERAN(v_GERANSysInfo24.Geran[0].BandIndicator), {}),

                               5),  // priority

      cs_CarrierFreqsInfoGERAN(cs_CarrierFreqsGERAN(v_GERANSysInfo25.Geran[0].Arfcn, f_ConvertGERAN_BandToBandIndicatorGERAN(v_GERANSysInfo25.Geran[0].BandIndicator), {}),

                               5)   // priority

    };

    v_SystemInformationBlockType7 := cs_508_SystemInformationBlockType7_Def(v_CarrierFreqsInfoListGERAN, 7); 

        f_EUTRA_CellInfo_SetSIB7(eutra_Cell1,

                             v_SystemInformationBlockType7);

     f_EUTRA_CellInfo_SetSIB7(eutra_Cell2,

                             v_SystemInformationBlockType7);

    //...the save back the SIB7

    f_EUTRA_CellInfo_SetSIB7(eutra_Cell1, v_SystemInformationBlockType7);

    f_EUTRA_CellInfo_SetSIB7(eutra_Cell2, v_SystemInformationBlockType7);
    //Create and configure Cells

    f_EUTRA_CellConfig_Def(eutra_Cell1);

    f_EUTRA_CellConfig_Def(eutra_Cell2);

    // Add the CSG ID of Cell 2 in CSG list in USIM

    // Preamble

    if (pc_Manual_CSG_Selection) { //@sic R5-134565 sic@

        // Insert a USIM

        f_UT_USIM_Insert (UT, "USIM with service n86 available");

        // Empty UE Allowed CSG list in USIM

        f_EUTRA_CellInfo_SetSysInfo_Csg_Ind(eutra_Cell2, true);

        f_EUTRA_EmptyACSGL ( eutra_Cell2, {v_CSG_Identity_Cell2}); //@sic R5-155978 sic@

        // Broadcast correct CSG id

        f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell2, v_CSG_Identity_Cell2);

        f_EUTRA_SS_ConfigureSysinfo(eutra_Cell2);

        // Add CSG2 in the Alllowed CSG list in USIM

        f_EUTRA_Manual_CSG_Selection( eutra_Cell2, v_CSG_Identity_Cell2);

    } else {

        // Insert a USIM UE CSG whitelist contains the CSG ID of Cell 2

        f_UT_USIM_Insert (UT, "USIM with service n86 available and UE CSG whitelist contains the CSG ID of Cell 2");

    }

    // Set Cell2 to Hybrid cell

    f_EUTRA_CellInfo_SetSysInfo_Csg_Ind(eutra_Cell2, false);

    f_EUTRA_SS_ConfigureSysinfo(eutra_Cell2);

    f_EUTRA_SetCellPowerList (v_CellPowerList_T0);

    f_EUTRA_TestBody_Set (true);

    // Anritsu_CSG_647 – Change 4.2 
    f_IRAT_SendCoOrd (GERAN, cms_IRAT_Trigger); 

    //wait for GERAN PTC to perform steps 1 and 2

    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(GERAN);

    //@siclog "Step 3" siclog@

    f_EUTRA_SetCellPowerList (v_CellPowerList_T0toT1);

    //@siclog "Step 4" siclog@

    // The UE is made to perform manual CSG ID selection and select Cell 2.

    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell2);

    v_PLMN  := f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity);

    f_UT_ManualPLMN_CSG ( UT, v_PLMN, v_CSG_Identity_Cell2);

    //@siclog "Step 5" siclog@

    //Check: Does the test result of generic test procedure in TS 36.508 Table 6.4.2.7A-2 indicate that

    //the UE is camped on E-GERAN Cell 2?

    //NOTE: The UE performs a TAU procedure and the RRC connection is released.

    // Anritsu_CSG_647 – Change 4.3
    // f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT (eutra_Cell2, v_CSInOtherRAT, false, true, omit);

    f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT (eutra_Cell2, v_CSInOtherRAT, false, false, omit); //@sic R5-123734 sic@

    //@siclog "Step 6" siclog@

    //The SS changes Cell 1, Cell 2, Cell 24 and Cell 25 levels according

    // to row "T2" in table 6.4.7.3.2-1 and table 6.4.7.3.2-2.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T1toT2);

    f_EUTRA_SendAuthParameters(eutra_Cell2, GERAN);

    f_IRAT_SendCoOrd (GERAN, cms_IRAT_OctetData (hex2oct(tsc_EpsDefaultBearerId))); // @sic R5s130883 sic@

    f_IRAT_WaitForCoOrd_Trigger (GERAN);

    //@siclog "Step 18" siclog@

    f_EUTRA_SetCellPowerList (v_CellPowerList_T2toT3);

    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(GERAN);

    //@siclog "Step 30" siclog@

    f_EUTRA_SetCellPowerList (v_CellPowerList_T3toT4);

    //@siclog "Step 31" siclog@

    //Check that UE camps on Cell 2 with the generic test procedure TS 36.508 table 6.4.2.7A-2

    // Anritsu_CSG_647 – Change 4.4
    //f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT (eutra_Cell2, v_CSInOtherRAT, false, true, omit);

    f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT (eutra_Cell2, v_CSInOtherRAT, false, false, omit); //@sic R5-123734 sic@
    f_EUTRA_TestBody_Set (false);

    // Tell GERAN to cells down as test now finished

    f_IRAT_SendCoOrd (GERAN, cms_IRAT_Trigger);

    //@sic R5-155978 sic@
    f_EUTRA_SetCellPowerList (v_CellPowerList_Postamble);

    // Anritsu_CSG_647 – Change 4.5
    f_EUTRA_Postamble_EmptyCSG_AllowedList (eutra_Cell2, eutra_Cell2, {v_CSG_Identity_Cell2}, E1_IDLE);
  } // end of 6.4.7

5. Execution Log Files

5.1 INTEL UE
INTEL 7360 UE passed this test case on the Anritsu ME7834L LTE system with NAS and AS Snow 3G integrity and ciphering algorithm. The documentation below is enclosed as evidence of the successful test case run [1]
· Test Case Execution log file :

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file

	
	


